

# HAPcryst

**A HAP extension for crystallographic  
groups**

Version 0.1.15

26 July 2022

**Marc Roeder**

**Marc Roeder**

Email: [roeder.marc@gmail.com](mailto:roeder.marc@gmail.com)

## Copyright

© 2007 Marc Röder.

This package is distributed under the terms of the GNU General Public License version 2 or later (at your convenience). See the file LICENSE or <https://www.gnu.org/copyleft/gpl.html>

## Acknowledgements

This work was supported by Marie Curie Grant No. MTKD-CT-2006-042685

# Contents

<b>1</b>	<b>Introduction</b>	<b>4</b>
1.1	Abstract and Notation . . . . .	4
1.2	Requirements . . . . .	5
1.3	Global Variables . . . . .	5
<b>2</b>	<b>Bits and Pieces</b>	<b>6</b>
2.1	Matrices and Vectors . . . . .	6
2.2	Affine Matrices OnRight . . . . .	7
2.3	Geometry . . . . .	8
2.4	Space Groups . . . . .	9
<b>3</b>	<b>Algorithms of Orbit-Stabilizer Type</b>	<b>10</b>
3.1	Orbit Stabilizer for Crystallographic Groups . . . . .	10
<b>4</b>	<b>Resolutions of Crystallographic Groups</b>	<b>15</b>
4.1	Fundamental Domains . . . . .	15
4.2	Face Lattice and Resolution . . . . .	18
	<b>References</b>	<b>21</b>
	<b>Index</b>	<b>22</b>

# Chapter 1

## Introduction

### 1.1 Abstract and Notation

HAPcryst is an extension for "Homological Algebra Programming" (HAP, [Ell]) by Graham Ellis. It uses geometric methods to calculate resolutions for crystallographic groups. In this manual, we will use the terms "space group" and "crystallographic group" synonymous. As usual in GAP, group elements are supposed to act from the right. To emphasize this fact, some functions have names ending in "OnRight" (namely those, which rely on the action from the right). This is also meant to make work with HAPcryst and cryst [EGN] easier.

The functions called "somethingStandardSpaceGroup" are supposed to work for standard crystallographic groups on left and right some time in the future. Currently only the versions acting on right are implemented. As in cryst [EGN], space groups are represented as affine linear groups. For the computations in HAPcryst, crystallographic groups have to be in "standard form". That is, the translation basis has to be the standard basis of the space. This implies that the linear part of a group element need not be orthogonal with respect to the usual scalar product.

#### 1.1.1 The natural action of crystallographic groups

There is some confusion about the way crystallographic groups are written. This concerns the question if we act on left or on right and if vectors are of the form  $[1, \dots]$  or  $[\dots, 1]$ .

As mentioned, HAPcryst handles affine crystallographic groups on right (and maybe later also on left) acting on vectors of the form  $[\dots, 1]$ .

BUT: The functions in HAPcryst do not take augmented vectors as input (no leading or ending ones). The handling of vectors is done internally. So in HAPcryst, a crystallographic group is a group of  $n \times n$  matrices which acts on a vector space of dimension  $n - 1$  whose elements are vectors of length  $n - 1$  (not  $n$ ). Example:

Example

```
gap> G:=SpaceGroup(3,4); #This group acts on 3-Space
SpaceGroupOnRightBBNWZ( 3, 2, 1, 1, 2 )
gap> Display(Representative(G));
[ [ 1, 0, 0, 0 ],
  [ 0, 1, 0, 0 ],
  [ 0, 0, 1, 0 ],
  [ 0, 0, 0, 1 ] ]
gap> OrbitStabilizerInUnitCubeOnRight(G, [1/2,0,0]);
rec( orbit := [ [ 1/2, 0, 0 ], [ 1/2, 1/2, 0 ] ],
```

```
stabilizer := Group([ [ [ 1, 0, 0, 0 ], [ 0, 1, 0, 0 ], [ 0, 0, 1, 0 ],
                        [ 0, 0, 0, 1 ] ] ] )
```

## 1.2 Requirements

The following GAP packages are required

- polymaking which in turn depends on the computational geometry software polymake.
- HAP
- Cryst

The following GAP packages are not required but highly recommended:

- carat
- CrystCat
- GAPDoc is needed to display the online manual

### 1.2.1 Recommendation concerning polymake

Calculating resolutions of Bieberbach groups involves convex hull computations. polymake by default uses cdd to compute convex hulls. Experiments suggest that lrs is the more suitable algorithm for the computations done in HAPcryst than the default cdd. You can change the behaviour of by editing the file "yourhomedirectory/.polymake/prefer.pl". It should contain a section like this (just make sure lrs is before cdd, the position of beneath\_beyond does not matter):

```
#####
application polytope;

prefer "*.convex_hull lrs, beneath_beyond, cdd";
```

## 1.3 Global Variables

HAPcryst itself does only have one global variable, namely InfoHAPcryst (1.3.1). The location of files generated for interaction with polymake are determined by the value of POLYMAKE\_DATA\_DIR (**polymaking: POLYMAKE\_DATA\_DIR**) which is a global variable of polymaking.

### 1.3.1 InfoHAPcryst

▷ InfoHAPcryst

(info class)

At a level of 1, only the most important messages are printed. At level 2, additional information is displayed, and level 3 is even more verbose. At level 0, HAPcryst remains silent.

## Chapter 2

# Bits and Pieces

This chapter contains a few very basic functions which are needed for space group calculations and were missing in standard GAP.

### 2.1 Matrices and Vectors

#### 2.1.1 SignRat

▷ `SignRat(x)` (method)

**Returns:** sign of the rational number  $x$  (Standard GAP currently only has `SignInt`).

#### 2.1.2 VectorModOne

▷ `VectorModOne(v)` (method)

**Returns:** Rational vector of the same length with entries in  $[0, 1)$

For a rational vector  $v$ , this returns the vector with all entries taken "mod 1".

Example

```
gap> SignRat((-4)/(-2));  
1  
gap> SignRat(9/(-2));  
-1  
gap> VectorModOne([1/10, 100/9, 5/6, 6/5]);  
[ 1/10, 1/9, 5/6, 1/5 ]
```

#### 2.1.3 IsSquareMat

▷ `IsSquareMat(matrix)` (method)

**Returns:** true if *matrix* is a square matrix and false otherwise.

#### 2.1.4 DimensionSquareMat

▷ `DimensionSquareMat(matrix)` (method)

**Returns:** Number of lines in the matrix *matrix* if it is square and fail otherwise

Example

```
gap> m:=[[1,2,3],[4,5,6],[9,6,12]];  
[ [ 1, 2, 3 ], [ 4, 5, 6 ], [ 9, 6, 12 ] ]
```

```
gap> IsSquareMat(m);
true
gap> DimensionSquareMat(m);
3
gap> DimensionSquareMat([[1,2],[1,2,3]]);
Error, Matrix is not square called from
```

Affine mappings of  $n$  dimensional space are often written as a pair  $(A, v)$  where  $A$  is a linear mapping and  $v$  is a vector. GAP represents affine mappings by  $n + 1$  times  $n + 1$  matrices  $M$  which satisfy  $M_{n+1,n+1} = 1$  and  $M_{i,n+1} = 0$  for all  $1 \leq i \leq n$ .

An affine matrix acts on an  $n$  dimensional space which is written as a space of  $n + 1$  tuples with  $n + 1$ st entry 1. Here we give two functions to handle these affine matrices.

## 2.2 Affine Matrices OnRight

### 2.2.1 LinearPartOfAffineMatOnRight

▷ `LinearPartOfAffineMatOnRight(mat)` (method)

**Returns:** the linear part of the affine matrix  $mat$ . That is, everything except for the last row and column.

### 2.2.2 BasisChangeAffineMatOnRight

▷ `BasisChangeAffineMatOnRight(transform, mat)` (method)

**Returns:** affine matrix with same dimensions as  $mat$

A basis change  $transform$  of an  $n$  dimensional space induces a transformation on affine mappings on this space. If  $mat$  is a affine matrix (in particular, it is  $(n + 1) \times (n + 1)$ ), this method returns the image of  $mat$  under the basis transformation induced by  $transform$ .

Example

```
gap> c:=[[0,1],[1,0]];
[ [ 0, 1 ], [ 1, 0 ] ]
gap> m:=[[1/2,0,0],[0,2/3,0],[1,0,1]];
[ [ 1/2, 0, 0 ], [ 0, 2/3, 0 ], [ 1, 0, 1 ] ]
gap> BasisChangeAffineMatOnRight(c,m);
[ [ 2/3, 0, 0 ], [ 0, 1/2, 0 ], [ 0, 1, 1 ] ]
```

### 2.2.3 TranslationOnRightFromVector

▷ `TranslationOnRightFromVector(v)` (method)

**Returns:** Affine matrix

Given a vector  $v$  with  $n$  entries, this method returns a  $(n + 1) \times (n + 1)$  matrix which corresponds to the affine translation defined by  $v$ .

Example

```
gap> m:=TranslationOnRightFromVector([1,2,3]);;
gap> Display(m);
[ [ 1, 0, 0, 0 ],
  [ 0, 1, 0, 0 ],
  [ 0, 0, 1, 0 ],
```

```

[ 1, 2, 3, 1 ] ]
gap> LinearPartOfAffineMatOnRight(m);
[ [ 1, 0, 0 ], [ 0, 1, 0 ], [ 0, 0, 1 ] ]
gap> BasisChangeAffineMatOnRight([ [3,2,1], [0,1,0], [0,0,1] ],m);
[ [ 1, 0, 0, 0 ], [ 0, 1, 0, 0 ], [ 0, 0, 1, 0 ], [ 3, 4, 4, 1 ] ]

```

## 2.3 Geometry

### 2.3.1 GramianOfAverageScalarProductFromFiniteMatrixGroup

▷ GramianOfAverageScalarProductFromFiniteMatrixGroup( $G$ ) (method)

**Returns:** Symmetric positive definite matrix

For a finite matrix group  $G$ , the gramian matrix of the average scalar product is returned. This is the sum over all  $gg^t$  with  $g \in G$  (actually it is enough to take a generating set). The group  $G$  is orthogonal with respect to the scalar product induced by the returned matrix.

### 2.3.2 Inequalities

Inequalities are represented in the same way they are represented in `polymaking`. The vector  $(v_0, \dots, v_n)$  represents the inequality  $0 \leq v_0 + v_1x_1 + \dots + v_nx_n$ .

### 2.3.3 BisectorInequalityFromPointPair

▷ BisectorInequalityFromPointPair( $v1$ ,  $v2$  [,  $gram$ ]) (method)

**Returns:** vector of length  $\text{Length}(v1)+1$

Calculates the inequality defining the half-space containing  $v1$  such that  $v1 - v2$  is perpendicular on the bounding hyperplane. And  $(v1 - v2)/2$  is contained in the bounding hyperplane.

If the matrix  $gram$  is given, it is used as the gramian matrix. Otherwise, the standard scalar product is used. It is not checked if  $gram$  is positive definite or symmetric.

### 2.3.4 WhichSideOfHyperplane

▷ WhichSideOfHyperplane( $v$ ,  $ineq$ ) (method)

▷ WhichSideOfHyperplaneNC( $v$ ,  $ineq$ ) (method)

**Returns:** -1 (below) 0 (in) or 1 (above).

Let  $v$  be a vector of length  $n$  and  $ineq$  an inequality represented by a vector of length  $n + 1$ . Then `WhichSideOfHyperplane( $v$ ,  $ineq$ )` returns 1 if  $v$  is a solution of the inequality but not the equation given by  $ineq$ , it returns 0 if  $v$  is a solution to the equation and -1 if it is not a solution of the inequality  $ineq$ .

The NC version does not test the input for correctness.

Example

```

gap> BisectorInequalityFromPointPair([0,0],[1,0]);
[ 1, -2, 0 ]
gap> ineq:=BisectorInequalityFromPointPair([0,0],[1,0],[[5,4],[4,5]]);
[ 5, -10, -8 ]
gap> ineq{[2,3]}*[1/2,0];
-5
gap> WhichSideOfHyperplane([0,0],ineq);

```



```

1
gap> WhichSideOfHyperplane([1/2,0],ineq);
0

```

### 2.3.5 RelativePositionPointAndPolygon

▷ RelativePositionPointAndPolygon(*point*, *poly*) (method)

**Returns:** one of "VERTEX", "FACET", "OUTSIDE", "INSIDE"

Let *poly* be a PolymakeObject and *point* a vector. If *point* is a vertex of *poly*, the string "VERTEX" is returned. If *point* lies inside *poly*, "INSIDE" is returned and if it lies in a facet, "FACET" is returned and if *point* does not lie inside *poly*, the function returns "OUTSIDE".

## 2.4 Space Groups

### 2.4.1 PointGroupRepresentatives

▷ PointGroupRepresentatives(*group*) (attribute)

▷ PointGroupRepresentatives(*group*) (method)

**Returns:** list of matrices

Given an AffineCrystGroupOnLeftOrRight *group*, this returns a list of representatives of the point group of *group*. That is, a system of representatives for the factor group modulo translations. This is an attribute of AffineCrystGroupOnLeftOrRight

## Chapter 3

# Algorithms of Orbit-Stabilizer Type

We introduce a way to calculate a sufficient part of an orbit and the stabilizer of a point.

### 3.1 Orbit Stabilizer for Crystallographic Groups

#### 3.1.1 OrbitStabilizerInUnitCubeOnRight

▷ `OrbitStabilizerInUnitCubeOnRight(group, x)` (method)

**Returns:** A record containing

- `.stabilizer`: the stabilizer of  $x$ .
- `.orbit` set of vectors from  $[0, 1)^n$  which represents the orbit.

Let  $x$  be a rational vector from  $[0, 1)^n$  and  $group$  a space group in standard form. The function then calculates the part of the orbit which lies inside the cube  $[0, 1)^n$  and the stabilizer of  $x$ . Observe that every element of the full orbit differs from a point in the returned orbit only by a pure translation.

Note that the restriction to points from  $[0, 1)^n$  makes sense if orbits should be compared and the vector passed to `OrbitStabilizerInUnitCubeOnRight` should be an element of the returned orbit (part).

Example

```
gap> S:=SpaceGroup(3,5);;
gap> OrbitStabilizerInUnitCubeOnRight(S,[1/2,0,9/11]);
rec( orbit := [ [ 0, 1/2, 2/11 ], [ 1/2, 0, 9/11 ] ],
      stabilizer := Group([ [ 1, 0, 0, 0 ], [ 0, 1, 0, 0 ], [ 0, 0, 1, 0 ],
                           [ 0, 0, 0, 1 ] ])) )
gap> OrbitStabilizerInUnitCubeOnRight(S,[0,0,0]);
rec( orbit := [ [ 0, 0, 0 ] ], stabilizer := <matrix group with 2 generators> )
```

If you are interested in other parts of the orbit, you can use `VectorModOne` (2.1.2) for the base point and the functions `ShiftedOrbitPart` (3.1.9), `TranslationsToOneCubeAroundCenter` (3.1.10) and `TranslationsToBox` (3.1.11) for the resulting orbit

Suppose we want to calculate the part of the orbit of  $[4/3, 5/3, 7/3]$  in the cube of sidelength 1 around this point:

## Example

```
gap> S:=SpaceGroup(3,5);;
gap> p:=[4/3,5/3,7/3];;
gap> o:=OrbitStabilizerInUnitCubeOnRight(S,VectorModOne(p)).orbit;
[ [ 1/3, 2/3, 1/3 ], [ 1/3, 2/3, 2/3 ] ]
gap> box:=p+[[ -1,1 ], [ -1,1 ], [ -1,1 ]];
[ [ 1/3, 8/3, 7/3 ], [ 1/3, 8/3, 7/3 ], [ 1/3, 8/3, 7/3 ] ]
gap> o2:=Concatenation(List(o,i->i+TranslationsToBox(i,box)));;
gap> # This is what we looked for. But it is somewhat large:
gap> Size(o2);
54
```

### 3.1.2 OrbitStabilizerInUnitCubeOnRightOnSets

▷ OrbitStabilizerInUnitCubeOnRightOnSets(*group*, *set*)

(method)

**Returns:** A record containing

- .stabilizer: the stabilizer of *set*.
- .orbit set of sets of vectors from  $[0,1)^n$  which represents the orbit.

Calculates orbit and stabilizer of a set of vectors. Just as OrbitStabilizerInUnitCubeOnRight (3.1.1), it needs input from  $[0,1)^n$ . The returned orbit part .orbit is a set of sets such that every element of .orbit has a non-trivial intersection with the cube  $[0,1)^n$ . In general, these sets will not lie inside  $[0,1)^n$  completely.

## Example

```
gap> S:=SpaceGroup(3,5);;
gap> OrbitStabilizerInUnitCubeOnRightOnSets(S,[[0,0,0],[0,1/2,0]]);
rec( orbit := [ [ [ -1/2, 0, 0 ], [ 0, 0, 0 ] ],
                [ [ 0, 0, 0 ], [ 0, 1/2, 0 ] ],
                [ [ 1/2, 0, 0 ], [ 1, 0, 0 ] ] ],
      stabilizer := Group([ [ [ 1, 0, 0, 0 ], [ 0, 1, 0, 0 ],
                             [ 0, 0, 1, 0 ], [ 0, 0, 0, 1 ] ] ])) )
```

### 3.1.3 OrbitPartInVertexSetsStandardSpaceGroup

▷ OrbitPartInVertexSetsStandardSpaceGroup(*group*, *vertexset*, *allvertices*) (method)

**Returns:** Set of subsets of *allvertices*.

If *allvertices* is a set of vectors and *vertexset* is a subset thereof, then OrbitPartInVertexSetsStandardSpaceGroup returns that part of the orbit of *vertexset* which consists entirely of subsets of *allvertices*. Note that, unlike the other OrbitStabilizer algorithms, this does not require the input to lie in some particular part of the space.

## Example

```
gap> S:=SpaceGroup(3,5);;
gap> OrbitPartInVertexSetsStandardSpaceGroup(S,[[0,1,5],[1,2,0]],
> Set([[1,2,0],[2,3,1],[1,2,6],[1,1,0],[0,1,5],[3/5,7,12],[1/17,6,1/2]]));
[ [ [ 0, 1, 5 ], [ 1, 2, 0 ] ], [ [ 1, 2, 6 ], [ 2, 3, 1 ] ] ]
gap> OrbitPartInVertexSetsStandardSpaceGroup(S, [[1,2,0]],
> Set([[1,2,0],[2,3,1],[1,2,6],[1,1,0],[0,1,5],[3/5,7,12],[1/17,6,1/2]]));
[ [ [ 0, 1, 5 ] ], [ [ 1, 1, 0 ] ], [ [ 1, 2, 0 ] ], [ [ 1, 2, 6 ] ], [ [ 2, 3, 1 ] ] ] ]
```

### 3.1.4 OrbitPartInFacesStandardSpaceGroup

▷ `OrbitPartInFacesStandardSpaceGroup(group, vertexset, faceset)` (method)

**Returns:** Set of subsets of *faceset*.

This calculates the orbit of a space group on sets restricted to a set of faces.

If *faceset* is a set of sets of vectors and *vertexset* is an element of *faceset*, then `OrbitPartInFacesStandardSpaceGroup` returns that part of the orbit of *vertexset* which consists entirely of elements of *faceset*.

Note that, unlike the other `OrbitStabilizer` algorithms, this does not require the input to lie in some particular part of the space.

### 3.1.5 OrbitPartAndRepresentativesInFacesStandardSpaceGroup

▷ `OrbitPartAndRepresentativesInFacesStandardSpaceGroup(group, vertexset, faceset)` (method)

**Returns:** A set of face-matrix pairs.

This is a slight variation of `OrbitPartInFacesStandardSpaceGroup` (3.1.4) that also returns a representative for every orbit element.

Example

```
gap> S:=SpaceGroup(3,5);;
gap> OrbitPartInVertexSetsStandardSpaceGroup(S, [[0,1,5],[1,2,0]],
> Set([[1,2,0],[2,3,1],[1,2,6],[1,1,0],[0,1,5],[3/5,7,12],[1/17,6,1/2]]));
[[ [ [ 0, 1, 5 ], [ 1, 2, 0 ] ], [ [ 1, 2, 6 ], [ 2, 3, 1 ] ] ]
gap> OrbitPartInFacesStandardSpaceGroup(S, [[0,1,5],[1,2,0]],
> Set( [ [ [ 0, 1, 5 ], [ 1, 2, 0 ] ], [[1/17,6,1/2],[1,2,7]] ]));
[[ [ [ 0, 1, 5 ], [ 1, 2, 0 ] ] ]
gap> OrbitPartAndRepresentativesInFacesStandardSpaceGroup(S, [[0,1,5],[1,2,0]],
> Set( [ [ [ 0, 1, 5 ], [ 1, 2, 0 ] ], [[1/17,6,1/2],[1,2,7]] ]));
[[ [ [ 0, 1, 5 ], [ 1, 2, 0 ] ],
[ [ 1, 0, 0, 0 ], [ 0, 1, 0, 0 ], [ 0, 0, 1, 0 ], [ 0, 0, 0, 1 ] ] ] ] ]
```

### 3.1.6 StabilizerOnSetsStandardSpaceGroup

▷ `StabilizerOnSetsStandardSpaceGroup(group, set)` (method)

**Returns:** finite group of affine matrices (OnRight)

Given a set *set* of vectors and a space group *group* in standard form, this method calculates the stabilizer of that set in the full crystallographic group.

Example

```
gap> G:=SpaceGroup(3,12);;
gap> v:=[ 0, 0,0 ];;
gap> s:=StabilizerOnSetsStandardSpaceGroup(G,[v]);
<matrix group with 2 generators>
gap> s2:=OrbitStabilizerInUnitCubeOnRight(G,v).stabilizer;
<matrix group with 2 generators>
gap> s2=s;
true
```

### 3.1.7 RepresentativeActionOnRightOnSets

▷ `RepresentativeActionOnRightOnSets(group, set, imageset)` (method)

**Returns:** Affine matrix.

Returns an element of the space group  $S$  which takes the set  $set$  to the set  $imageset$ . The group must be in standard form and act on the right.

Example

```
gap> S:=SpaceGroup(3,5);;
gap> RepresentativeActionOnRightOnSets(G, [[0,0,0],[0,1/2,0]],
>      [[ 0, 1/2, 0 ], [ 0, 1, 0 ] ]);
[ [ 0, -1, 0, 0 ], [ -1, 0, 0, 0 ], [ 0, 0, -1, 0 ], [ 0, 1, 0, 1 ] ]
```

### 3.1.8 Getting other orbit parts

HAPcryst does not calculate the full orbit but only the part of it having coefficients between  $-1/2$  and  $1/2$ . The other parts of the orbit can be calculated using the following functions.

### 3.1.9 ShiftedOrbitPart

▷ `ShiftedOrbitPart(point, orbitpart)` (method)

**Returns:** Set of vectors

Takes each vector in  $orbitpart$  to the cube unit cube centered in  $point$ .

Example

```
gap> ShiftedOrbitPart([0,0,0],[[1/2,1/2,1/3],[-1/2,1/2,1/2],[19,3,1]]);
[ [ 1/2, 1/2, 1/3 ], [ 1/2, 1/2, 1/2 ], [ 0, 0, 0 ] ]
gap> ShiftedOrbitPart([1,1,1],[[1/2,1/2,1/2],[-1/2,1/2,1/2]]);
[ [ 3/2, 3/2, 3/2 ] ]
```

### 3.1.10 TranslationsToOneCubeAroundCenter

▷ `TranslationsToOneCubeAroundCenter(point, center)` (method)

**Returns:** List of integer vectors

This method returns the list of all integer vectors which translate  $point$  into the box  $center+[-1/2,1/2]^n$

Example

```
gap> TranslationsToOneCubeAroundCenter([1/2,1/2,1/3],[0,0,0]);
[ [ 0, 0, 0 ], [ 0, -1, 0 ], [ -1, 0, 0 ], [ -1, -1, 0 ] ]
gap> TranslationsToOneCubeAroundCenter([1,0,1],[0,0,0]);
[ [ -1, 0, -1 ] ]
```

### 3.1.11 TranslationsToBox

▷ `TranslationsToBox(point, box)` (method)

**Returns:** An iterator of integer vectors or the empty iterator

Given a vector  $v$  and a list of pairs, this function returns the translation vectors (integer vectors) which take  $v$  into the box  $box$ . The box  $box$  has to be given as a list of pairs.

## Example

```
gap> TranslationsToBox([0,0],[[1/2,2/3],[1/2,2/3]]);  
[ ]  
gap> TranslationsToBox([0,0],[[-3/2,1/2],[1,4/3]]);  
[ [ -1, 1 ], [ 0, 1 ] ]  
gap> TranslationsToBox([0,0],[[-3/2,1/2],[2,1]]);  
Error, Box must not be empty called from  
...
```

## Chapter 4

# Resolutions of Crystallographic Groups

### 4.1 Fundamental Domains

Let  $S$  be a crystallographic group. A Fundamental domain is a closed convex set containing a system of representatives for the Orbits of  $S$  in its natural action on euclidian space.

There are two algorithms for calculating fundamental domains in HAPcryst. One uses the geometry and relies on having the standard rule for evaluating the scalar product (i.e. the gramian matrix is the identity). The other one is independent of the gramian matrix but does only work for Bieberbach groups, while the first ("geometric") algorithm works for arbitrary crystallographic groups given a point with trivial stabilizer.

#### 4.1.1 FundamentalDomainStandardSpaceGroup

- ▷ `FundamentalDomainStandardSpaceGroup([v, ]G)` (method)
- ▷ `FundamentalDomainStandardSpaceGroup(v, G)` (method)

**Returns:** a PolymakeObject

Let  $G$  be an `AffineCrystGroupOnRight` and  $v$  a vector. A fundamental domain containing  $v$  is calculated and returned as a `PolymakeObject`. The vector  $v$  is used as the starting point for a Dirichlet-Voronoi construction. If no  $v$  is given, the origin is used as starting point if it has trivial stabiliser. Otherwise an error is cast.

Example

```
gap> fd:=FundamentalDomainStandardSpaceGroup([1/2,0,1/5],SpaceGroup(3,9));
<polymake object>
gap> Polymake(fd,"N_VERTICES");
24
gap> fd:=FundamentalDomainStandardSpaceGroup(SpaceGroup(3,9));
<polymake object>
gap> Polymake(fd,"N_VERTICES");
8
```

#### 4.1.2 FundamentalDomainBieberbachGroup

- ▷ `FundamentalDomainBieberbachGroup(G)` (method)
- ▷ `FundamentalDomainBieberbachGroup(v, G[, gram])` (method)

**Returns:** a PolymakeObject

Given a starting vector  $v$  and a Bieberbach group  $G$  in standard form, this method calculates the Dirichlet domain with respect to  $v$ . If *gram* is not supplied, the average gramian matrix is used (see `GramianOfAverageScalarProductFromFiniteMatrixGroup` (2.3.1)). It is not tested if *gram* is symmetric and positive definite. It is also not tested, if the product defined by *gram* is invariant under the point group of  $G$ .

The behaviour of this function is influenced by the option `ineqThreshold`. The algorithm calculates approximations to a fundamental domain by iteratively adding inequalities. For an approximating polyhedron, every vertex is tested to find new inequalities. When all vertices have been considered or the number of new inequalities already found exceeds the value of `ineqThreshold`, a new approximating polyhedron is calculated. The default for `ineqThreshold` is 200. Roughly speaking, a large threshold means shifting work from `polymake` to `GAP`, a small one means more calls of (and work for) `polymake`.

If the value of `InfoHAPcryst` (1.3.1) is 2 or more, for each approximation the number of vertices of the approximation, the number of vertices that have to be considered during the calculation, the number of facets, and new inequalities is shown.

Note that the algorithm chooses vertices in random order and also writes inequalities for `polymake` in random order.

#### Example

```
gap> a0:=[[ 1, 0, 0, 0, 0, 0, 0, 0 ], [ 0, -1, 0, 0, 0, 0, 0, 0 ],
>      [ 0, 0, 1, 0, 0, 0, 0, 0 ], [ 0, 0, 0, 1, 0, 0, 0, 0 ],
>      [ 0, 0, 0, 0, 0, 1, 0, 0 ], [ 0, 0, 0, 0, 0, -1, -1, 0 ],
>      [ -1/2, 0, 0, 1/6, 0, 0, 0, 1 ]
>      ];;
gap> a1:=[[ 0, -1, 0, 0, 0, 0, 0, 0 ], [ 0, 0, -1, 0, 0, 0, 0, 0 ],
>      [ 1, 0, 0, 0, 0, 0, 0, 0 ], [ 0, 0, 0, 1, 0, 0, 0, 0 ],
>      [ 0, 0, 0, 0, 1, 0, 0, 0 ], [ 0, 0, 0, 0, 0, 1, 0, 0 ],
>      [ 0, 0, 0, 0, 1/3, -1/3, 1 ]
>      ];;
gap> trans:=List(Group((1,2,3,4,5,6)),g->
>      TranslationOnRightFromVector(Permuted([1,0,0,0,0,0],g)));
gap> S:=AffineCrytGroupOnRight(Concatenation(trans,[a0,a1]));
<matrix group with 8 generators>
gap> SetInfoLevel(InfoHAPcryst,2);
gap> FundamentalDomainBieberbachGroup(S:ineqThreshold:=10);
#I v: 104/104 f:15
#I new: 201
#I v: 961/961 f:58
#I new: 20
#I v: 1143/805 f:69
#I new: 12
#I v: 1059/555 f:64
#I new: 15
#I v: 328/109 f:33
#I new: 12
#I v: 336/58 f:32
#I new: 0
<polymake object>
gap> FundamentalDomainBieberbachGroup(S:ineqThreshold:=1000);
#I v: 104/104 f:15
#I new: 149
#I v: 635/635 f:41
```



```
#I new: 115
#I v: 336/183 f:32
#I new: 0
#I out of inequalities
<polymake object>
```

### 4.1.3 FundamentalDomainFromGeneralPointAndOrbitPartGeometric

▷ `FundamentalDomainFromGeneralPointAndOrbitPartGeometric(v, orbit)` (method)

**Returns:** a `PolymakeObject`

This uses an alternative algorithm based on geometric considerations. It is not used in any of the high-level methods. Let  $v$  be a vector and  $orbit$  a sufficiently large part of the orbit of  $v$  under a crystallographic group with standard-orthogonal point group (satisfying  $A^t = A^{-1}$ ). A geometric algorithm is then used to calculate the Dirichlet domain with respect to  $v$ . This also works for crystallographic groups which are not Bieberbach. The point  $v$  has to have trivial stabilizer.

The intersection of the full orbit with the unit cube around  $v$  is sufficiently large.

Example

```
gap> G:=SpaceGroup(3,9);;
gap> v:=[0,0,0];
[ 0, 0, 0 ]
gap> orbit:=OrbitStabilizerInUnitCubeOnRight(G,v).orbit;
[ [ 0, 0, 0 ], [ 0, 0, 1/2 ] ]
gap> fd:=FundamentalDomainFromGeneralPointAndOrbitPartGeometric(v,orbit);
<polymake object>
gap> Polymake(fd,"N_VERTICES");
8
```

### 4.1.4 IsFundamentalDomainStandardSpaceGroup

▷ `IsFundamentalDomainStandardSpaceGroup(poly, G)` (method)

**Returns:** true or false

This tests if a `PolymakeObject`  $poly$  is a fundamental domain for the affine crystallographic group  $G$  in standard form.

The function tests the following: First, does the orbit of any vertex of  $poly$  have a point inside  $poly$  (if this is the case, false is returned). Second: Is every facet of  $poly$  the image of a different facet under a group element which does not fix  $poly$ . If this is satisfied, true is returned.

### 4.1.5 IsFundamentalDomainBieberbachGroup

▷ `IsFundamentalDomainBieberbachGroup(poly, G)` (method)

**Returns:** true, false or fail

This tests if a `PolymakeObject`  $poly$  is a fundamental domain for the affine crystallographic group  $G$  in standard form and if this group is torsion free (ie a Bieberbach group)

It returns true if  $G$  is torsion free and  $poly$  is a fundamental domain for  $G$ . If  $poly$  is not a fundamental domain, false is returned regardless of the structure of  $G$ . And if  $G$  is not torsion free, the method returns fail. If  $G$  is polycyclic, torsion freeness is tested using a representation as pcg group. Otherwise the stabilisers of the faces of the fundamental domain  $poly$  are calculated ( $G$  is torsion free if and only if all these stabilisers are trivial).

## 4.2 Face Lattice and Resolution

For Bieberbach groups (torsion free crystallographic groups), the following functions calculate free resolutions. This calculation is done by finding a fundamental domain for the group. For a description of the HapResolution datatype, see the Hap data types documentation or the experimental datatypes documentation **HAPprog: Resolutions in Hap**

### 4.2.1 ResolutionBieberbachGroup

▷ ResolutionBieberbachGroup( $G$  [,  $v$ ]) (method)

**Returns:** a HAPresolution

Let  $G$  be a Bieberbach group given as an AffineCrystGroupOnRight and  $v$  a vector. Then a Dirichlet domain with respect to  $v$  is calculated using FundamentalDomainBieberbachGroup (4.1.2). From this domain, a resolution is calculated using FaceLatticeAndBoundaryBieberbachGroup (4.2.2) and ResolutionFromFLandBoundary (4.2.3). If  $v$  is not given, the origin is used.

Example

```
gap> R:=ResolutionBieberbachGroup(SpaceGroup(3,9));
Resolution of length 3 in characteristic
0 for SpaceGroupOnRightBBNWZ( 3, 2, 2, 2, 2 ) .
No contracting homotopy available.

gap> List([0..3],Dimension(R));
[ 1, 3, 3, 1 ]
gap> R:=ResolutionBieberbachGroup(SpaceGroup(3,9),[1/2,0,0]);
Resolution of length 3 in characteristic
0 for SpaceGroupOnRightBBNWZ( 3, 2, 2, 2, 2 ) .
No contracting homotopy available.

gap> List([0..3],Dimension(R));
[ 6, 12, 7, 1 ]
```

### 4.2.2 FaceLatticeAndBoundaryBieberbachGroup

▷ FaceLatticeAndBoundaryBieberbachGroup( $poly$ ,  $group$ ) (method)

**Returns:** Record with entries .hasse and .elts representing a part of the hasse diagram and a lookup table of group elements

Let  $group$  be a torsion free AffineCrystGroupOnRight (that is, a Bieberbach group). Given a PolymakeObject  $poly$  representing a fundamental domain for  $group$ , this method uses polymaking to calculate the face lattice of  $poly$ . From the set of faces, a system of representatives for  $group$ -orbits is chosen. For each representative, the boundary is then calculated. The list .elts contains elements of  $group$  (in fact, it is even a set). The structure of the returned list .hasse is as follows:

- The  $i$ -th entry contains a system of representatives for the  $i - 1$  dimensional faces of  $poly$ .
- Each face is represented by a pair of lists [vertices,boundary]. The list of integers vertices represents the vertices of  $poly$  which are contained in this face. The enumeration is chosen such that an  $i$  in the list represents the  $i$ -th entry of the list Polymake(poly,"VERTICES");

- The list boundary represents the boundary of the respective face. It is a list of pairs of integers  $[j, g]$ . The first entry lies between  $-n$  and  $n$ , where  $n$  is the number of faces of dimension  $i - 1$ . This entry represents a face of dimension  $i - 1$  (or its additive inverse as a module generator). The second entry  $g$  is the position of the matrix in `.elts`.

This representation is compatible with the representation of free  $\mathbb{Z}G$  modules in `Hap` and this method essentially calculates a free resolution of `group`. If the value of `InfoHAPcryst` (1.3.1) is 2 or more, additional information about the number of faces in every codimension, the number of orbits of the group on the free module generated by those faces, and the time it took to calculate the orbit decomposition is output.

Example

```
gap> SetInfoLevel(InfoHAPcryst,2);
gap> G:=SpaceGroup(3,165);
SpaceGroupOnRightBBNWZ( 3, 6, 1, 1, 4 )
gap> fd:=FundamentalDomainBieberbachGroup(G);
<polymake object>
gap> fl:=FaceLatticeAndBoundaryBieberbachGroup(fd,G);
#I 1(4/8): 0:00:00.004
#I 2(5/18): 0:00:00.000
#I 3(2/12): 0:00:00.000
#I Face lattice done ( 0:00:00.004). Calculating boundary
#I done ( 0:00:00.004) Reformating...
gap> RecNames(fl);
[ "hasse", "elts", "grouping" ]
gap> fl.grouping;
<free left module over Integers, and ring-with-one, with 10 generators>
```

### 4.2.3 ResolutionFromFLandBoundary

▷ `ResolutionFromFLandBoundary(fl, group)`

(method)

**Returns:** Free resolution

If `fl` is the record output by `FaceLatticeAndBoundaryBieberbachGroup` (4.2.2) and `group` is the corresponding group, this function returns a `HapResolution`. Of course, `fl` has to be generated from a fundamental domain for `group`

Example

```
gap> G:=SpaceGroup(3,165);
SpaceGroupOnRightBBNWZ( 3, 6, 1, 1, 4 )
gap> fd:=FundamentalDomainBieberbachGroup(G);
<polymake object>
gap> fl:=FaceLatticeAndBoundaryBieberbachGroup(fd,G);
gap> ResolutionFromFLandBoundary(fl,G);
Resolution of length 3 in characteristic
0 for SpaceGroupOnRightBBNWZ( 3, 6, 1, 1, 4 ) .
No contracting homotopy available.

gap> ResolutionFromFLandBoundary(fl,G);
Resolution of length 3 in characteristic
0 for SpaceGroupOnRightBBNWZ( 3, 6, 1, 1, 4 ) .
No contracting homotopy available.
```

```
gap> List([0..4],Dimension(last));  
[ 2, 5, 4, 1, 0 ]
```

# References

- [EGN] Bettina Eick, Franz Gahler, and Werner Nickel. `cryst`. <https://www.gap-system.org/Packages/cryst.html>. 4
- [Ell] Graham Ellis. `Hap`. <http://hamilton.nuigalway.ie/Hap/www/>. 4

# Index

action of crystallographic groups, 4

BasisChangeAffineMatOnRight, 7

BisectorInequalityFromPointPair, 8

DimensionSquareMat, 6

FaceLatticeAndBoundaryBieberbachGroup, 18

FundamentalDomainBieberbachGroup, 15

FundamentalDomainFromGeneralPointAndOrbitPartGeometric, 17

FundamentalDomainStandardSpaceGroup, 15

GramianOfAverageScalarProductFromFiniteMatrixGroup, 8

ineqThreshold, 16

InfoHAPcryst, 5

installation, 5

IsFundamentalDomainBieberbachGroup, 17

IsFundamentalDomainStandardSpaceGroup, 17

IsSquareMat, 6

LinearPartOfAffineMatOnRight, 7

OrbitPartAndRepresentativesInFacesStandardSpaceGroup, 12

OrbitPartInFacesStandardSpaceGroup, 12

OrbitPartInVertexSetsStandardSpaceGroup, 11

OrbitStabilizerInUnitCubeOnRight, 10

OrbitStabilizerInUnitCubeOnRightOnSets, 11

PointGroupRepresentatives, 9

polymake, 5

RelativePositionPointAndPolygon, 9

RepresentativeActionOnRightOnSets, 13

ResolutionBieberbachGroup, 18

ResolutionFromFLandBoundary, 19

ShiftedOrbitPart, 13

SignRat, 6

StabilizerOnSetsStandardSpaceGroup, 12

TranslationOnRightFromVector, 7

TranslationsToBox, 13

TranslationsToOneCubeAroundCenter, 13

VectorModOne, 6

WhichSideOfHyperplane, 8

WhichSideOfHyperplaneNC, 8